Pangram checker

You are encouraged to solve this task according to the task description, using any language you may know.
A pangram is a sentence that contains all the letters of the English alphabet at least once.

For example:   The quick brown fox jumps over the lazy dog.



Task

Write a function or method to check a sentence to see if it is a   pangram   (or not)   and show its use.



Related tasks

   determine if a string has all the same characters
   determine if a string has all unique characters


$ include "seed7_05.s7i";
 
const func boolean: isPangram (in string: stri) is func
  result
    var boolean: isPangram is FALSE;
  local
    var char: ch is ' ';
    var set of char: usedChars is (set of char).value;
  begin
    for ch range lower(stri) do
      if ch in {'a' .. 'z'} then
        incl(usedChars, ch);
      end if;
    end for;
    isPangram := usedChars = {'a' .. 'z'};
  end func;
 
const proc: main is func
  begin
    writeln(isPangram("This is a test"));
    writeln(isPangram("The quick brown fox jumps over the lazy dog"));
    writeln(isPangram("NOPQRSTUVWXYZ  abcdefghijklm"));
    writeln(isPangram("abcdefghijklopqrstuvwxyz"));  # Missing m, n
  end func;